

Rec'd. 10/3/84

USING ATARI TOUCH TABLET WITH BASIC

- 1) Embedding an AtariArtist picture in a BASIC program
- 2) Using the ATARI Touch Tablet as an input device

ATARI, INC.
CONSUMER PRODUCT SERVICE
PRODUCT SUPPORT GROUP
1312 Crossman Avenue
Sunnyvale, CA 94088

~~(800) 672-1404 inside CA~~
~~(800) 538-6543 outside CA~~

408 - 745 - 4851

DEMOPAC #12
Rev. 1 2-84 JGS

JTB

Using ATARI Touch Tablet with BASIC

Introduction

JQS 2/14/84

There are two ways to use the ATARI Touch Tablet within a BASIC program. 1) you can create a picture with the AtariArtist cartridge and then, by saving it with a special procedure not documented in the manual you can "call" the file into a BASIC program; 2) you can create pictures with your BASIC cartridge using the Touch Tablet as an input device, much as you would a paddle or a joystick. The obvious advantage of the first procedure is the use of the fabulous AtariArtist cartridge. The disadvantage is that you cannot use this procedure on a cassette. The second procedure can be saved to either disk or cassette.

I. EMBEDDING AN ATARIARTIST PICTURE IN A BASIC PROGRAM - DISK USE ONLY!

Many thanks to John Clark at Atari, Inc. for creating the following program.

1. Using the ATARI Touch Tablet and the AtariArtist cartridge create a picture of your own.
2. With the picture still on the screen simply press the INSERT key. Your picture will be saved in an uncondensed form, taking up 62 sectors. (Using the usual storage procedure as described in the Touch Tablet manual, a picture is condensed for the convenience of conserving disk space. The condensing procedure means the file cannot be read by any other language than the AtariArtist cartridge. Atari, Inc. will soon release materials which will allow you to access this form with BASIC, but in the meantime you may use the program in this DEMOPAC.)
3. Using the INSERT key, the picture will always be saved by the filename PICTURE. If you save a second picture on the same disk it will replace the first file, causing you to lose the first picture. To avoid this, simply go to the DOS menu (by removing the AtariArtist cartridge), choose option E. RENAME FILE, and follow the procedure outlined in the DOS II manual, page 32. If you have DOS III, use the equivalent function.
4. When you have saved the pictures you want on the disk, leave the picture disk in the disk drive and the disk drive on, remove the AtariArtist cartridge, insert the BASIC cartridge and turn on the computer.
5. Type in your BASIC program, following the procedure outlined in the program following. In line 120, change the filename to whatever filename you have. If you have several pictures, you can call them each by name. The pictures will appear in Atari's default colors (orange, green, blue). To change the colors please refer to the Color List following.

COLOR LIST

The pictures "called" by this procedure will always be in Atari's default colors orange, green and blue. You may change these to whatever you like using the following information.

The colors are put into registers much as paint is put into paint buckets. To change colors you need to tell the computer three things: 1) which register you want, 2) which color you want and 3) how bright you want the color. You use the command SETCOLOR (which may be abbreviated to SE.) followed by three numbers.

SE. X, Y, Z

where X is the register number,
Y is the color number, and
Z is the luminance number (brightness).

REGISTER NUMBER

When you have loaded your picture and it displays the three colors orange, green and blue,

if the color is ORANGE the register number is 0
if the color is GREEN the register number is 1
if the color is BLUE the register number is 2.

COLOR NUMBER

0 = GRAY	8 = ANOTHER BLUE
1 = GOLD	9 = LIGHT BLUE
2 = ORANGE	10 = TURQUOISE
3 = RED-ORANGE	11 = GREEN-BLUE
4 = PINK	12 = GREEN
5 = PINK-PURPLE	13 = YELLOW-GREEN
6 = PURPLE-BLUE	14 = ORANGE-GREEN
7 = BLUE	15 = LIGHT ORANGE

LUMINANCE (BRIGHTNESS) NUMBER

0, 2, 4, 6, 8, 10, 12, 14
darker -- lighter

EXAMPLE: insert a line 36 to read:

36 SE.0,3,2:SE.1,2,10:SE.2,10,6

This line could be added to either program and would change the orange to a dark red-orange, the green to a light orange and the blue to a medium turquoise.

ADDITIONAL INSTRUCTIONS FOR YOUR BASIC PROGRAMS

1. AtariArtist creates pictures in ANTIC mode 14 (also called 7 1/2 since it is between Graphics 7 and 8 in resolution). This mode is not accessible to BASIC in the Atari 400 or 800 but is in the XL series of ATARI computers. This is why there are two different listings.
2. Line 150 indicates where your BASIC program would continue. If you have the Atari 400 or 800, you will need to state a new Graphics mode, since the one this program uses will not be able to "understand" graphics commands from BASIC on these machines. Simply make one of the first lines following 150 state GRAPHICS 0 (or whatever mode number you want).
3. If you have an Atari XL computer, you will be "in" GRAPHICS 15. This can be addressed in BASIC on these machines, so you only need to use a GRAPHICS command if you want to change modes.

PLEASE NOTE: The BASIC programs on this page use a machine language program called by the USR command in line 140. The machine language program itself is in the form of numbers in lines 505 - 515. For technically advanced programmers we have included the machine language program on the next two pages.

For the ATARI 400 or 800:

```
10 REM      **** Artshow ****
11 REM      * by John Clark *
12 REM      ***** 2/9/84 ****
13 REM
20 DIM FILE$(15)
30 GRAPHICS 8+16
35 REM Set up a Graphics "7 1/2" (ANTIC mode 14) display list
40 DL=PEEK(560)+PEEK(561)*256
50 GOSUB 500:REM Poke assembly language routine into memory
55 REM Now change the Gr. 8 display list to "7 1/2"
60 POKE DL+3,PEEK(DL+3)-1
70 FOR I=6 TO 98
80 POKE DL+I,14:NEXT I
90 POKE DL+99,PEEK(DL+99)-1
100 FOR I=102 TO 198
110 POKE DL+I,14:NEXT I
120 FILE$="D:PICTURE":REM or put your filespec here
130 BUFADR=PEEK(88)+PEEK(89)*256:REM Screen
140 A=USR(1536,ADR(FILE$),4,0,7680,7,BUFADR)
150 GOTO 150:REM or continue with your BASIC program
500 FOR I=0 TO 65:READ J:POKE 1536+I,J:NEXT I:RETURN
505 DATA 104,160,3,140,82,3,104,141,85,3,104,141,84,3,104,104,141,90,3,104,104,141,91,3,162,16,32,86,228,104
510 DATA 141,89,3,104,141,88,3,104,104,141,82,3,104,141,85,3,104,141,84,3,162,16,32,86,228,169,12,141,82,3
515 DATA 162,16,32,86,228,96
```

For the ATARI 1200XL, 600XL (with additional memory to support a disk drive) and 800XL:

```
10 REM      **** Artshow ****
11 REM      * by John Clark *
12 REM      ***** 2/9/84 ****
13 REM
20 DIM FILE$(15)
30 GRAPHICS 15+16
40 GOSUB 500:REM Poke assembly language routine into memory
50 FILE$="D:PICTURE":REM or put your filespec here
60 BUFADR=PEEK(88)+PEEK(89)*256:REM Screen
70 A=USR(1536,ADR(FILE$),4,0,7680,7,BUFADR)
80 GOTO 150:REM or continue with your BASIC program
500 FOR I=0 TO 65:READ J:POKE 1536+I,J:NEXT I:RETURN
505 DATA 104,160,3,140,82,3,104,141,85,3,104,141,84,3,104,104,141,90,3,104,104,141,91,3,162,16,32,86,228,104
510 DATA 141,89,3,104,141,88,3,104,104,141,82,3,104,141,85,3,104,141,84,3,162,16,32,86,228,169,12,141,82,3
515 DATA 162,16,32,86,228,96
```


II. USING THE ATARI TOUCH TABLET AS AN INPUT DEVICE

Many thanks to Diane Heininger and Bill Carris, both of Atari, Inc., for developing the following programs.

The positioning of the ATARI Touch Tablet's screen is quite different from the normal graphics modes of BASIC. Position 0,0 is not at the upper left of the screen, as in BASIC, but at the lower left. It is designed like a standard two dimensional graph. The horizontal axis (the X axis) extends from position 0 at the extreme left to 227 at the extreme right. The vertical axis (the Y axis) extends from 0 position at the bottom to 227 at the top.

The Touch Tablet is an "input" device for the computer, providing a means of telling the computer to perform a certain function. With the ATARI Touch Tablet, you can do this: (1) by touching the stylus (or other object like your finger) to the surface of the tablet and drawing, (2) by pressing the left button, (3) by pressing the right button, or (4) by pressing the red button on the stylus. The computer distinguishes these signals in the following ways...

1. DRAWING ON THE TABLET SURFACE

PADDLE(0) reads the value of the X axis. Type in and RUN the following program and move the stylus randomly over the tablet surface.

```
10 PRINT PADDLE(0)
20 GOTO 10
```

You will see that the numbers of 0 for extreme left and 227 for extreme right are ideal numbers. The real range may not be that great. It also makes a difference how fast you move. If no pressure is applied to the surface, the reading is 228.

Similarly, PADDLE(1) reads the value of the Y axis. Change the above program to PADDLE(1). You will observe the numbers change from near 0 at the bottom of the tablet to near 228 at the top. When no pressure is applied, the reading will be 228.

2. LEFT BUTTON

The button on the top left side of the tablet is recognized by PTRIG(0). This variable returns a value of 0 when pressed and 1 when not pressed. Use the following program to demonstrate this.

```
10 PRINT PTRIG(0)
20 GOTO 20
```

3. RIGHT BUTTON

The button on the right side of the tablet is recognized by PTRIG(1). Modify the above program to PTRIG(1) and see how it works.

4. RED BUTTON ON STYLUS

The red button on the drawing stylus is recognized by STICK(0). When the button is pressed STICK(0)=14. Use this program to show this:

```
10 PRINT STICK(0)
20 GOTO 20
```

With the AtariArtist cartridge, pressing the three buttons will perform various functions, such as creating a continuous drawing line, going from one screen to another, etc. With your own BASIC program, you can choose to have the buttons do different things in conjunction with moving the stylus. For instance, you might use the red button to draw in one color and the right or left buttons to draw in a different color....but there are many more possibilities.

Screen Graphics and the Touch Tablet

As stated above, the Touch Tablet is divided both vertically and horizontally in units from 0 to 227. This does not correspond proportionally to any of the ATARI graphics modes. The following information is provided to help you write Touch Tablet programs which use Graphics 7 or 8.

GRAPHICS 7

The resolution of Graphics 7 is 160 pixels across and 80 scan lines down with a text window, and 96 down without. The resolution of the tablet is greater than this in both directions. In order to make the tablet correspond directly to the dimensions of Graphics 7, two simple fractions are used:

For the X axis - $159/227$ - in the programs that fraction is referred to as XLIMIT.

For the Y axis - $79/227$ (or $95/227$ w/o text window) - this is called YLIMIT.

GRAPHICS 8

The Graphics 8 resolution is 320 across by 160 down with text window and 192 down without. To make the tablet correspond to this, use

$319/227$ for XLIMIT

$159/227$ ($191/227$ w/o text window) for YLIMIT

SCREEN X,Y COORDINATES, TOUCH TABLET COORDINATES

The Graphics Modes also differ in another way from the layout of the ATARI Touch Tablet, as has been mentioned earlier. The Origin point of the tablet is in the lower left corner. In the Graphics Modes, however, the origin point is in the upper left corner of the screen. This difference doesn't affect the X axis at all, but it does have an impact on the Y axis. The following equation will convert the Y axis properly:

$(YAXIS-228)*-1$ equals the vertical point on the tablet

Omitting this equation will make your picture upside down!

The following programs are some suggestion to direct you to use of the tablet with BASIC. You will get different results with these programs when you move the stylus slowly than when you move quickly. The REM statements show you what each program is designed to do.

```

5 REM Touch.BAS
6 REM This program will make dots if you move fast and lines if you move
7 REM slowly. Because of the fine resolution, the
8 REM different colors do not show up very well.
9 REM by DH & BC
10 PRINT CHR$(125):REM Clears the screen
20 GRAPHICS 7+16
30 XAXIS=PADDLE(0)
40 YAXIS=PADDLE(1)
50 XLIMIT=0.697
55 YLIMIT=0.417
59 REM Flashing cursor and eraser
60 COLOR X:PLOT XAXIS*XLIMIT,(YAXIS-228)*-YLIMIT:FOR DE=1 TO 2:NEXT DE
70 COLOR 4:PLOT XAXIS*XLIMIT,(YAXIS-228)*-YLIMIT:FOR DE=1 TO 2:NEXT DE
79 REM Red button
80 IF STICK(0)=14 THEN GOTO 200
89 REM Left button
90 IF PTRIG(0)=0 THEN GOTO 300
99 REM Right button
100 IF PTRIG(1)=0 THEN GOTO 400
110 GOTO 30
200 X=1:GOTO 500
300 X=2:GOTO 500
400 X=3:GOTO 500
500 IF XAXIS>227 THEN GOTO 30
510 IF YAXIS>227 THEN GOTO 30
520 PLOT XAXIS*XLIMIT,(YAXIS-228)*-YLIMIT
530 GOTO 30

```

```

5 REM Soundtou.ch
6 REM This program is similar to Touch.BAS, except
7 REM it will make a sound when the buttons are pressed.
8 REM by DH & BC
10 PRINT CHR$(125):REM Clears the screen
20 GRAPHICS 7+16
30 XAXIS=PADDLE(0)
40 YAXIS=PADDLE(1)
50 XLIMIT=0.697
55 YLIMIT=0.417
59 REM Flashing cursor and eraser
60 COLOR X:PLOT XAXIS*XLIMIT,(YAXIS-228)*-YLIMIT:FOR DE=1 TO 2:NEXT DE
70 COLOR 4:PLOT XAXIS*XLIMIT,(YAXIS-228)*-YLIMIT:FOR DE=1 TO 2:NEXT DE
79 REM Red button
80 IF STICK(0)=14 THEN GOTO 200
89 REM Left button
90 IF PTRIG(0)=0 THEN GOTO 300
99 REM Right button
100 IF PTRIG(1)=0 THEN GOTO 400
110 GOTO 30
200 X=1:GOTO 500
300 X=2:GOTO 500
400 X=3:GOTO 500
500 IF XAXIS>227 THEN GOTO 30
510 IF YAXIS>227 THEN GOTO 30
520 SOUND 0,INT(XAXIS*XLIMIT),10,10:SOUND 1,INT((YAXIS-228)*-YLIMIT),10,10
525 IF STICK(0)<>14 THEN SOUND 0,0,0,0:SOUND 1,0,0,0
530 GOTO 30

```

```

5 PRINT CHR$(125):REM Basic7.TT
6 REM The line appears when the
7 REM left and right buttons are pushed.
8 REM by DH & BC
10 GRAPHICS 7+16
20 XAXIS=PADDLE(0)
30 YAXIS=PADDLE(1)
40 IF PTRIG(0)=0 THEN GOTO 100
50 IF PTRIG(1)=0 THEN GOTO 70
60 GOTO 20
70 COLOR 2:SETCOLOR 1,5,6
80 GOTO 300
100 COLOR 1:SETCOLOR 0,0,15
300 PLOT XAXIS*0.697,(YAXIS-228)*-0.417
310 GOTO 20

```

```

5 PRINT CHR$(125):REM Basic8.TT
6 REM Just a simple program by DH & BC
10 GRAPHICS 8+16:COLOR 1:SETCOLOR 1,0,0:SETCOLOR 2,0,15
20 XAXIS=PADDLE(0)
30 YAXIS=PADDLE(1)
40 IF XAXIS>227 THEN XAXIS=227:GOTO 20
50 IF YAXIS>191 THEN YAXIS=191:GOTO 20
60 PLOT XAXIS*1.4,(YAXIS-228)*-0.84
70 GOTO 20

```

```

5 PRINT CHR$(125):REM Stick7
6 REM The line appears when the stylus
7 REM and right buttons are pushed.
8 REM by DH & BC
10 GRAPHICS 7+16
20 XAXIS=PADDLE(0)
30 YAXIS=PADDLE(1)
40 IF STICK(0)=14 THEN GOTO 100
50 IF PTRIG(1)=0 THEN GOTO 70
60 GOTO 20
70 COLOR 2:SETCOLOR 1,5,6
80 GOTO 300
100 COLOR 1:SETCOLOR 0,0,15
300 PLOT XAXIS*0.697,(YAXIS-228)*-0.417
310 GOTO 20

```

```

5 REM Tablet1.BAS
6 REM This program uses the buttons
7 REM to clear the screen after drawing.
8 REM by DH & BC 1/84
10 GRAPHICS 7+16
20 COLOR 1
30 X=PADDLE(0)
40 Y=PADDLE(1)
45 YFLIP=-Y+228
50 L=PTRIG(0)
60 R=PTRIG(1)
70 RED=STICK(0)
80 XLIMIT=159/227
90 YLIMIT=95/227
100 IF X>159 THEN X=159
110 IF Y>95 THEN Y=95
120 IF R=0 OR L=0 OR RED=14 THEN GOTO 500
130 PLOT X*XLIMIT,YLIMIT*YFLIP
140 GOTO 30
500 GOTO 10

```

```

5 REM Tablet2.BAS
6 REM How to plot upside down!
7 REM by DH & BC
10 GRAPHICS 7+16
20 COLOR 1
100 X=PADDLE(0)
110 Y=PADDLE(1)
170 IF X>159 THEN X=159
180 IF Y>95 THEN Y=95
200 PLOT X,Y
210 GOTO 100

```

```

5 REM Tablet3.BAS
6 REM Just another way of drawing
7 REM by DH & BC
10 GRAPHICS 7+16
20 COLOR 1
100 X=PADDLE(0)
110 Y=PADDLE(1)
120 YFLIP=-Y+228
150 XLIMIT=159/227
160 YLIMIT=95/227
170 IF X>227 THEN X=227
180 IF Y>227 THEN Y=227
200 PLOT X*XLIMIT,YLIMIT*YFLIP
210 GOTO 100

```

```

5 REM ChngColr.BAS
6 REM This program uses the buttons
7 REM to change colors.
8 REM No color will show until a
9 REM button is pressed. by JQS 2/84
10 GRAPHICS 7+16
30 X=PADDLE(0)
40 Y=PADDLE(1)
45 YFLIP=-Y+228
50 L=PTRIG(0)
60 R=PTRIG(1)
70 RED=STICK(0)
80 XLIMIT=159/227
90 YLIMIT=95/227
100 IF X>227 THEN X=227
110 IF Y>227 THEN Y=227
120 IF L=0 THEN C=2
121 IF R=0 THEN C=3
122 IF RED=14 THEN C=1
130 COLOR C:PLOT X*XLIMIT,YLIMIT*YFLIP
140 GOTO 30

```

```

5 REM Lines.TT
6 REM Press the stylus button to
7 REM start a line. Press the left
8 REM button and a straight line will be
9 REM formed between the two points.
10 REM by JQS, 2/84
20 GRAPHICS 7+16
30 COLOR 1
40 X=PADDLE(0)
50 Y=PADDLE(1)
60 XLIMIT=159/227
70 YLIMIT=95/227
80 YFLIP=-Y+228
90 IF X>227 THEN X=227
100 IF Y>227 THEN Y=227
200 IF STICK(0)=14 THEN PLOT X*XLIMIT,YLIMIT*YFLIP
210 FOR W=1 TO 20:NEXT W
220 IF PTRIG(0)=0 THEN DRAWTO X*XLIMIT,YLIMIT*YFLIP
230 GOTO 30

```